

OBSERVATORIO NACIONAL
-CENTRO DE ESTUDIOS-
REGISTRO NÚMERO CATALOGO
N.º ARMARIO TABLA

EFECTOS DE LA ATMOSFERA TERRESTRE
SOBRE LA PROPAGACION DE
ONDAS ELECTROMAGNETICAS

Juan Eusebio Garrido Arenas
Jesús Martín-Pintado

Informe Técnico C.A.Y. 1990 - 4

INDICE

1.-	Introducción	2
2.-	Primera modificación	4
3.-	Segunda modificación :	
	Modelo de Amósfera Estándar	4
4.-	Fiabilidad del Modelo de Atmósfera Estándar	8
5.-	Aplicación de cada opción	10
6.-	Tercera modificación :	
	Modelo de retardo introducido por la atmósfera	11
7.-	Cuarta modificación :	
	Dependencia con la elevación	14
ANEXOS:		
8.-	Tablas	17
9.-	Bibliografía	24
10.-	Listado del programa	29

MODIFICACIONES AL PROGRAMA INTERS

1. INTRODUCCION

El programa fuente original se llama INTERS. El lenguaje fuente usado es FORTRAN correspondiente al ordenador HP 1000 de Hewlett-Packard.

El objetivo del programa es calcular la cantidad de agua precipitable presente en la atmósfera, así como las emisividades y opacidades cenitales atmosféricas para un intervalo de frecuencias seleccionado por el usuario. Los datos de entrada corresponden a medidas de radiosondeos efectuados desde la estación meteorológica del Aeropuerto de Barajas: presión en milibares, temperatura en grados centígrados y humedad relativa en tanto por ciento o punto de rocío en grados centígrados, estos datos pueden ser introducidos por teclado o leídos de disco.

Un pseudocódigo no detallado podría ser el siguiente:

1. Preguntar opciones al usuario y pedir datos.

Leer opciones y datos.

- Integración: fina o gruesa
- Frecuencias: inicial, final e incremento
- ¿Procesar más datos?: Sí: seguir
No: parar
- ¿Datos leídos desde el terminal?: Sí o No
Sí: - fecha y hora
 - número de puntos notables
 - ¿humedad relativa o punto de rocío?

- para cada punto notable:

- presión

- temperatura

- humedad relativa o punto de rocío

- número de orden del punto notable equivocado

No: - nombre del fichero de datos

- número de días concretos

2. Transformación de punto de rocío en humedad relativa.

3. Interpolación lineal entre puntos notables cada 8 mb.

- corrección de datos de presión erróneos

- definición de punto notable a 10 mb

- interpolación:

- altura

- temperatura

- presión parcial de vapor de agua

4. Integración de la cantidad de agua precipitable y de la opacidad y emisividad atmosférica.

5. Presentación de resultados:

- cantidad de agua precipitable

- para cada frecuencia seleccionada:

- frecuencia

- opacidades:

- vapor de agua

- oxígeno

- total

- emisividad

2. PRIMERA MODIFICACION

Consiste en adaptar este programa fuente, cuya versión ejecutable ha sido compilada y funciona en el ordenador HP 1000, a un lenguaje FORTRAN 77 estándar de modo que pueda ser compilado y ejecutado en el mayor número posible de máquinas. En concreto interesa su funcionamiento en un entorno VAX-VMS.

También se incorpora la creación de un fichero de datos cuyo nombre se pregunta al usuario y donde se almacenan los datos y resultados más relevantes de cada ejecución.

Además, se han introducido mejoras de cara al diálogo usuario - programa y se ha añadido la presentación de un nuevo resultado: temperatura de brillo observable

3. SEGUNDA MODIFICACION: MODELO DE ATMOSFERA ESTANDAR

Consiste en incorporar al programa otra manera de rellenar los vectores de datos: presión, temperatura y humedad relativa. De este modo tenemos tres opciones que son ofrecidas al usuario mediante un menú para que éste seleccione la elegida.

Las dos primeras son las ya existentes: datos de radiosondeos por teclado o leídos del disco, la tercera es la nueva: ATMOSFERA ESTANDAR.

En esta última consideramos datos hasta una altitud de 24320 metros: dividimos en 39 intervalos de 600 m a partir de la altura de Yebes

sobre el nivel del mar (920 m). La presión en el último punto, es decir, a una altitud de 24320 m, es de 15,43 mb (con la hipótesis de atmósfera estándar), valor menor que 20 mb, para esta condición el programa crea un punto notable más a una presión de 10 mb, al cual corresponde una altitud de 25903 m, equivaliendo así la integración a la efectuada para radiosondeos.

La caracterización de cada punto es la siguiente:

1. Presión:

$$p(z) = 1013,2 (1 - 0.0065 \cdot z(m) / 288)^{5.2568} \quad (\text{mb})$$

2. Temperatura:

$$t(z) = 15 - 0.0065 \cdot z(m) \quad , z \leq 11000 \text{ m}$$

$$t(z) = -56.5 \quad , z > 11000 \text{ m} \quad (^\circ\text{C})$$

Al usuario se le pregunta la temperatura leída en superficie: TS y para hacer compatibles la fórmula del estándar con la leída, se modifica la ordenada en el origen, conservando la pendiente de la recta:

$$t(z) = 15 - 0.0065 \cdot z(m) + (TS - 9.02) \quad , z \leq 11000 \text{ m}$$

$$t(z) = -56.5 + (TS - 9.02) \quad , z > 11000 \text{ m} \quad (^\circ\text{C})$$

9.02 grados centígrados es la temperatura que correspondería a Yebes por la recta estándar, dada su altura.

3. Humedad relativa:

Hay dos opciones, presentadas al usuario:

1. Teclar temperatura: TS (°C) y humedad relativa: HS (%) en superficie.

Se calcula la humedad en tanto por uno y la temperatura en kelvins:

$$HSUP = HS/100$$

$$TSUP = TS+273 \quad (K)$$

Se calcula la presión parcial de vapor de agua:

$$ESUP = HSUP \cdot ESAT(TSUP) \quad (mb)$$

donde ESAT(TSUP) es la presión parcial de saturación del vapor de agua a temperatura TSUP y se evalúa usando la función PREH2O, perteneciente al programa fuente .

A partir de los valores anteriores calculamos la humedad absoluta en superficie:

$$HABS = HUMABS(TSUP, ESUP) \cdot \exp(ALT YE(m)/2000) \quad (g/m^3)$$

donde HUMABS es la humedad absoluta en g/m^3 correspondiente a TSUP y ESUP (en Yebes), ALTYE es la altura de Yebes sobre el nivel del mar, en metros. HUMABS se evalúa mediante la función del mismo nombre, existente en el programa fuente.

Calculamos la cantidad de agua precipitable, en mm:

$$WPREC = HABS \cdot H_0(km) \cdot \exp(-ALTYE(m)/H_0(m))$$

escogemos como altura característica de la exponencial, $H_0 = 2000$ m, según el INFORME TECNICO C.A.Y. 1987-5 del Centro Astronómico de Yebes.

2. Teclar la temperatura (°C) en superficie únicamente. Se calcula la cantidad de agua precipitable mediante la siguiente fórmula

empírica, deducida a partir de la expuesta por el Dr. Martín-Pintado en su Tesis de Licenciatura (1979), adaptada para datos de radiosondeos:

$$WPREC = 3,3 + 0,4 \cdot TS \quad (\text{mm})$$

a partir de ella se calcula la humedad absoluta en superficie:

$$HABS = (WPREC/H_0(\text{km})) \cdot \exp(\text{ALTYE}(\text{m})/2000) \quad (\text{g/m}^3)$$

con igual significado que en el apartado 1.

Caracterizamos la humedad absoluta como una exponencial en función de la altura:

$$HABS(z) = HABS \cdot \exp(-z(\text{m})/H_0(\text{m}))$$

De la humedad absoluta obtenemos la presión parcial de vapor de agua, en mb, para cada altura o punto del vector de datos:

$$E(i) = HABS(i) \cdot T(i)/216,9$$

y de ésta, la humedad relativa, en tanto por ciento:

$$DN(i) = E(i)/ESAT(T(i)) \cdot 100$$

con el significado de las variables como indicamos anteriormente.

Con lo anterior hemos obtenido 3 vectores de 40 puntos:

1. Temperatura (C)
2. Presión (mb)
3. Humedad relativa (%)

correspondiente, cada punto, a una altura, éstas son equiespaciadas con un intervalo de 600 m. Así queda modelada una atmósfera estándar correspondiente a Yebes, con la temperatura en superficie leída o con ésta y la humedad relativa en superficie, a partir de la cual se

obtendrán los datos sobre opacidad y emisividad cenitales, a las frecuencias seleccionadas por el usuario.

4. FIABILIDAD DEL MODELO DE ATMOSFERA ESTANDAR

Para comparar nuestro modelo de atmósfera estándar con la atmósfera real sobre el observatorio, hemos escogido al azar 30 radiosondeos correspondientes a días claros y 10 correspondientes a días nublados en el período comprendido entre Junio de 1977 y Noviembre de 1978, realizados desde la estación meteorológica del Aeropuerto de Barajas a las 12 horas T.U.

En cada radiosondeo se da la serie de puntos notables medidos, determinados por discontinuidades en los gradientes, bien de temperatura, bien de humedad relativa o punto de rocío. Los datos de cada punto notable son: presión, temperatura, humedad relativa y punto de rocío, en el programa se da la opción de introducir humedad relativa o punto de rocío ya que ambos a la vez son redundantes. Para este estudio hemos usado la humedad relativa ya que, además, el programa transforma los puntos de rocío, si han sido introducidos éstos, en humedad relativa.

Empleamos integración gruesa, por no diferir significativamente los resultados de los obtenidos con integración fina, y las frecuencias elegidas fueron entre 25 y 45 GHz con intervalos de 5 GHz. Los resultados, almacenados en ficheros, representan un interesante banco de datos. Presentamos al final unas tablas con estos resultados.

Posteriormente pasamos a ejecutar el programa en la opción de

atmósfera estándar para cada una de las dos opciones incluidas en aquélla.

En la opción de temperatura y humedad relativa en superficie, se introdujeron estos parámetros obtenidos de los valores iniciales de integración de cada radiosondeo, que representan puntos muy próximos a la altura de Yebes. En la opción de temperatura en superficie únicamente, este dato se obtuvo análogamente.

Los parámetros de comparación son la cantidad de agua precipitable obtenida tras integrar y la atenuación total cenital. Los errores obtenidos son:

	Cantidad de agua precipitable		Atenuación total	
	Tsup	Tsup y HRsup	Tsup	T y HR
Días claros	+ 8 %	+ 41 %	- 8 %	+ 8 %
Días nublados	- 43 %	+ 7 %	- 27 %	- 1 %

Estos valores tienen el significado siguiente: por ejemplo el + 8 % en la cantidad de agua precipitable, Tsup, días claros, quiere decir que de los 30 radiosondeos para días claros, las diferencias de cantidad de agua precipitable entre cada ejecución con atmósfera estándar y el radiosondeo correspondiente, promediadas con su signo, revelan un 8 % de divergencia sobre los valores obtenidos con atmósfera estándar. Aunque el número de casos es mayor cuando la atmósfera estándar da

valor menor, cuando da mayor, lo hace con valores porcentuales mayores sobre el valor del radiosondeo y la media resultante de error es mayor.

5. APLICACION DE CADA OPCION

De los datos anteriores es evidente que para días claros conviene usar la opción de temperatura en superficie únicamente y para días nublados es más exacta la opción de temperatura y humedad relativa en superficie.

Parece razonable ya que, en un día claro, la cantidad de agua precipitable tiende a ser menor que en un día nublado en la misma época del año y, además, la humedad relativa en superficie no es significativa ya que es variable y poco correlacionada con los valores en alturas superiores.

En un día nublado, sin embargo, la cantidad de agua en capas superiores es mayor y los valores de humedad relativa en superficie tienden a ser mayores que los que proporcionaría el modelo de temperatura únicamente, ideado para días claros, con lo cual, tener en cuenta este dato hace los valores de la cantidad de agua precipitable más acordes con los obtenidos para casos reales (radiosondeos).

6. TERCERA MODIFICACION: MODELO DE RETARDO INTRODUCIDO POR LA ATMOSFERA

En todo lo visto hasta ahora hemos tenido en cuenta únicamente el efecto de la atmósfera sobre la amplitud de la onda que la atraviesa, este efecto es una atenuación. También se ha obtenido la temperatura de emisión de la atmósfera y la temperatura de brillo observable desde la antena, que será ruido externo captado por la antena.

Ahora vamos a estudiar el efecto de la atmósfera sobre la fase de la onda, este efecto es un retardo. Hasta hace relativamente poco tiempo no había estudios disponibles sobre este tema, debido al escaso interés que tiene conocer el retardo que introduce la atmósfera en las aplicaciones de comunicaciones y observaciones radioastronómicas usuales; pero tras el desarrollo de las comunicaciones por satélite y de la interferometría de muy larga línea de base (VLBI), y del perfeccionamiento de los osciladores usados en los equipos, se ha llegado a la conclusión de que el retardo atmosférico introduce más error en la medida de fases, que el ruido de fase propio de los osciladores.

Concretamente, en las observaciones de VLBI se usa un maser de hidrógeno como reloj, con un ruido de fase mínimo, por ello conviene estimar el error de fase introducido por la atmósfera y compensarlo luego, al hacer los cálculos de correlación de las señales de los distintos radiotelescopios. El parámetro de interés es, pues, el retardo de fase diferencial entre las antenas que observan, que cambia

con las diferentes condiciones climatológicas de los lugares y las diferentes elevaciones de las antenas en un instante de la observación.

El retardo lo podemos dividir en varias componentes:

1. Retardo hidrostático
2. Retardo debido al vapor de agua
3. Retardo debido a la ionosfera

Para una excelente descripción de los dos primeros, ver la referencia 13, *Appendix A: Zenith delay formulas.*

El retardo hidrostático es función de la presión en superficie en el lugar de la observación y lo suponemos, por tanto, de lenta y poco significativa variación con el tiempo.

El retardo debido al vapor de agua es el más interesante de cara a determinar el retardo diferencial, ya que es el más variable con el tiempo. Para una determinación precisa, se usa un método basado en medidas de temperatura de brillo de la atmósfera efectuadas con radiómetros de vapor de agua (WVR). En este programa haremos el cálculo de este retardo integrando la expresión oportuna, obtenida de los perfiles de atmósfera, entre la altura de Yebes y la máxima del perfil atmosférico: 25903 m, cuando empleamos el modelo de atmósfera estándar, otro valor cercano cuando introducimos datos de radiosondeos.

El retardo ionosférico es difícil de determinar, por depender del perfil de la concentración de electrones, pero su variación no es

significativa a frecuencias mayores que 2 GHz, ya que es inversamente proporcional al cuadrado de la frecuencia.

En el programa proporcionamos una estimación del máximo retardo probable introducido por la ionosfera.

Fórmulas utilizadas

Concretando lo expresado anteriormente, las expresiones usadas son:

1. Retardo hidrostático

$$\Delta L_h \text{ (cm)} = 0.2277 \frac{P_o}{1 - 0.00266 \cdot \cos 2\Phi - 0.00028 \cdot H}$$

donde:

P_o : presión en superficie (mb)

Φ : latitud geográfica de la antena

H : altitud de la antena sobre el geoide (km)

2. Retardo debido al vapor de agua

$$\Delta L_w \text{ (cm)} = 37.7 \cdot \int_0^{\infty} \frac{e(h)}{T^2(h)} dh$$

donde:

$e(h)$: presión parcial de vapor de agua (mb)

$T(h)$: temperatura absoluta (K)

h : altura sobre el geoide (m)

$$\infty \cong 25903 \text{ m}$$

3. Retardo ionosférico

$$\Delta L_i \text{ (cm)} = \frac{3900}{f^2}$$

donde:

f : frecuencia (GHz)

esta expresión deriva de la siguiente:

$$\Delta \tau_i \text{ (seg)} = \frac{1.3 \cdot 10^{-7} \cdot N_T}{f^2}$$

donde:

N_T : integral de la concentración de electrones a lo largo del camino recorrido (el/m²)

f: frecuencia (Hz)

La N_T máxima para 30° de elevación es, aproximadamente, $2 \cdot 10^{18}$ el/m², considerando un modelo de ionosfera estratificada plana y distribución de electrones uniforme con la altura, tendríamos, en dirección al zenit, $N_{TZ} = 10^{18}$ el/m².

El cálculo lo hacemos para todas las frecuencias solicitadas por el usuario.

7. CUARTA MODIFICACION: DEPENDENCIA CON LA ELEVACION

Hasta ahora hemos considerado únicamente que observábamos en dirección al zenit, ahora vamos a introducir la dependencia con la elevación, de los parámetros descritos anteriormente, y que consistirá en multiplicarlos por un factor dependiente de la elevación y, en su caso de parámetros atmosféricos, conocido en la literatura como "mapping function".

Nosotros vamos a usar dos funciones de mapping, una de ellas, considerando un modelo de atmósfera estratificada plana, es suficiente para la atenuación en función de la elevación, cuando ésta es mayor que 5° y, aunque no sabemos, para el retardo ionosférico, hasta que elevación mínima es válida, suponemos el mismo rango que para la atenuación.

La segunda función considera una atmósfera estratificada esférica y un modelo de atmósfera muy similar al usado en nuestro modelo de atmósfera estándar, quedando el factor en función de la elevación y parámetros meteorológicos en superficie, por lo que se puede considerar válida para nuestro modelo y también para radiosondeos, ya que aquél se ajusta razonablemente a éstos.

Fórmulas utilizadas

1. Atmósfera plana

$$M = \frac{1}{\text{sen}(\varepsilon)}$$

donde:

ε : elevación

esta es la función más sencilla posible.

2. Atmósfera esférica

Cuando la elevación está entre 5 y 20 grados:

$$M = \frac{1}{\text{sen}(\varepsilon) + \frac{a}{\tan(\varepsilon) + \frac{b}{\text{sen}(\varepsilon) + c}}}$$

Cuando la elevación está entre 20 y 90 grados:

$$M = \frac{1}{\operatorname{sen}(\varepsilon) + \frac{a}{\operatorname{sen}(\varepsilon) + \frac{b}{\operatorname{sen}(\varepsilon) + c}}}$$

Los coeficientes son:

$$a = 0.001185 \cdot [1 + 0.6071 \cdot 10^{-4} \cdot (P_o - 1000) - \\ - 0.1471 \cdot 10^{-3} \cdot e_o + \\ + 0.3072 \cdot 10^{-2} \cdot (T_o - 293)]$$

$$b = 0.001144 \cdot [1 + 0.1164 \cdot 10^{-4} \cdot (P_o - 1000) + \\ + 0.2795 \cdot 10^{-3} \cdot e_o + \\ + 0.3109 \cdot 10^{-2} \cdot (T_o - 293)]$$

$$c = -0.0090$$

donde:

P_o : presión en superficie (mb)

e_o : presión parcial de vapor de agua (mb)

T_o : temperatura en superficie (K)

Hemos supuesto el coeficiente de variación de la temperatura con la altura, $\beta = -6.5$ °C/km y la altura de la tropopausa, $h_t = 11231$ m

8. Tablas

A continuación mostramos las tablas con los resultados obtenidos para las diferentes opciones de ejecución:

1. Datos de radiosondeos
2. Datos de atmósfera estándar
 - 2.1. Temperatura en superficie
 - 2.2. Temperatura y humedad relativa en superficie

En las tablas usamos los números anteriores como claves.

Tabla 1 y Tabla 2 recogen resultados de radiosondeos realizados en días claros.

En Tabla 1 se muestran como datos la temperatura y humedad relativa en superficie, obtenidas de cada radiosondeo y usadas para las ejecuciones con las opciones de atmósfera estándar, y como resultado la cantidad de agua precipitable.

En Tabla 2 se muestran los resultados de atenuación cenital atmosférica y temperatura de emisividad de la atmósfera.

Tabla 3 y Tabla 4 son análogas pero para días nublados.

Para días claros los radiosondeos del 1 al 10 están realizados en verano, del 11 al 21 en otoño, del 22 al 27 en invierno y del 28 al 30 en primavera. Los días nublados no están agrupados por estaciones, por ser pequeña la muestra.

TABLA 1

Días claros

Cantidad de agua precipitable (mm)

	Tra. sup. (C)	H. rel. sup. (%)	1.	2.1.	2.2.
1.	16.31	43.93	7.5	10.2	12.7
2.	17.09	67.05	15.7	10.5	20.3
3.	20.68	45.36	16.1	12.2	17.3
4.	24.33	37.28	19.2	14.0	17.8
5.	26.30	30.90	14.4	15.0	16.7
6.	24.01	46.34	22.9	13.8	21.7
7.	22.26	34.11	9.2	13.0	14.4
8.	19.60	34.01	11.6	11.5	12.1
9.	16.32	52.64	9.6	10.2	15.2
10.	15.27	26.09	4.3	9.7	7.0
11.	11.91	68.22	10.1	8.2	14.7
12.	13.49	40.13	11.0	8.9	9.6
13.	4.58	60.96	6.2	5.1	7.9
14.	7.29	77.97	9.4	6.2	12.2
15.	14.89	61.33	10.8	9.5	16.1
16.	13.83	47.43	9.0	9.0	11.6
17.	15.43	38.69	11.8	9.8	10.6
18.	13.21	45.78	8.3	8.8	10.8
19.	6.89	79.32	6.4	6.0	12.1

20.	8.03	70.84	11.5	6.5	11.7
21.	7.85	82.80	9.4	6.4	13.5
22.	8.88	37.35	3.5	6.9	6.6
23.	5.28	67.99	3.7	5.3	9.3
24.	5.76	59.26	9.1	5.5	8.4
25.	3.15	61.97	5.6	4.5	7.3
26.	5.42	28.79	2.3	5.4	4.0
27.	14.09	46.95	8.7	9.2	11.7
28.	11.12	64.73	7.8	7.8	13.2
29.	16.16	35.57	3.7	10.1	10.2
30.	17.23	54.95	17.2	10.6	16,8

TABLA 2

Días claros

Opacidad: τ (nepers) y emisividad: T (K) cenitales

	τ			T		
	1.	2.1.	2.2.	1.	2.1.	2.2.
1.	0.032	0.035	0.040	268.1	269.4	270.4
2.	0.047	0.036	0.053	275.3	270.4	273.2
3.	0.047	0.038	0.046	275.0	274.6	276.1
4.	0.050	0.040	0.046	278.8	278.9	279.9
5.	0.043	0.041	0.044	279.3	281.2	281.6
6.	0.057	0.040	0.053	281.7	278.5	280.4
7.	0.038	0.039	0.041	281.2	276.5	276.9
8.	0.039	0.035	0.038	280.2	269.8	273.5
9.	0.037	0.035	0.044	274.6	269.4	271.2
10.	0.028	0.035	0.030	267.6	268.2	266.8
11.	0.039	0.033	0.044	269.6	264.1	266.6
12.	0.039	0.034	0.035	269.9	266.0	266.3
13.	0.031	0.028	0.034	259.8	254.6	256.5
14.	0.038	0.030	0.041	264.0	258.2	261.2
15.	0.039	0.034	0.046	268.8	267.7	270.0
16.	0.036	0.034	0.038	273.5	266.4	267.5
17.	0.040	0.035	0.036	273.0	268.4	268.7
18.	0.034	0.033	0.037	267.8	265.7	266.6

19.	0.033	0.030	0.041	263.5	257.7	260.7
20.	0.042	0.031	0.040	264.5	259.2	261.7
21.	0.040	0.030	0.043	268.4	258.9	262.2
22.	0.028	0.031	0.030	263.8	260.3	260.1
23.	0.027	0.029	0.036	256.7	255.6	257.9
24.	0.037	0.029	0.034	262.5	256.2	258.0
25.	0.029	0.028	0.033	257.6	252.8	254.7
26.	0.024	0.029	0.026	257.9	255.8	254.7
27.	0.035	0.034	0.038	270.0	266.7	267.8
28.	0.036	0.032	0.042	265.9	263.1	265.4
29.	0.027	0.035	0.035	260.8	269.3	269.3
30.	0.051	0.036	0.047	274.3	270.5	272.5

TABLA 3

Días nublados

Cantidad de agua precipitable (mm)

	Tra. sup. (C)	H. rel. sup (%)	1.	2.1.	2.2.
1.	11.87	71.95	17.3	8.2	15.5
2.	9.88	81.79	14.8	7.3	15.4
3.	5.31	88.22	10.3	5.4	12.0
4.	5.74	71.81	12.6	5.5	10.1
5.	11.54	82.55	13.4	8.0	17.3
6.	8.84	70.76	9.9	6.9	12.4
7.	7.01	75.12	11.3	6.1	11.6
8.	9.12	77.56	11.2	7.0	13.8
9.	10.79	70.60	14.6	7.7	14.1
10.	11.44	53.21	10.9	8.0	11.1

TABLA 4

Días nublados

Opacidad: τ (nepers) y emisividad: T (K) cenitales

	τ			T		
	1.	2.1.	2.2.	1.	2.1.	2.2.
1.	0.054	0.033	0.046	266.9	264.0	266.8
2.	0.046	0.032	0.046	268.3	261.5	264.8
3.	0.041	0.029	0.041	257.8	255.7	259.2
4.	0.043	0.029	0.038	261.3	256.2	258.8
5.	0.044	0.032	0.049	268.2	263.6	266.9
6.	0.037	0.031	0.041	262.9	260.2	262.8
7.	0.041	0.030	0.040	260.7	257.9	260.7
8.	0.040	0.031	0.044	262.6	266.6	263.6
9.	0.046	0.032	0.044	268.8	262.7	265.3
10.	0.040	0.032	0.038	265.6	263.5	264.9

8. BIBLIOGRAFIA

- [1] P. Planesas, L. A. Delgado, M. Serra, F. Pérez Murano y M. P. Zugasti.

"Parámetros atmosféricos medidos y determinados en el C. A. Y. desde Marzo de 1982 hasta Junio de 1987"

Informe Técnico del C. A. Y. 1987-5

- [2] P. Planesas, M. P. Zugasti

"Sobre la calibración de espectros en la banda de 40 a 50 GHz en el C. A. Y."

Informe Técnico del C. A. Y. 1987-6

- [3] D. Fernández Illán

"Caracterización de la atmósfera de Yebes para su utilización en la calibración de radioobservaciones."

Tesis de Licenciatura. Facultad de Ciencias Físicas. Universidad Complutense. Madrid

- [4] Roger G. Barry, Richard G. Chorley

"Atmósfera, tiempo y clima", 4 edición

Ed. Omega, 1985, Barcelona

- [5] Sverre Pettersen, Ph. D.
"Introducción a la meteorología", 5 edición
Ed. Espasa-Calpe, S.A., Madrid, 1976
- [6] D. H. McIntosh, A. S. Thom
"Meteorología básica"
Ed. Alhambra, 1983
- [7] J. M. Jansa Guardiola
"Manual del observador de meteorología"
Instituto Nacional de Meteorología, Madrid, 1968
- [8] E. Bollay, N. R. Beers
"Handbook of meteorology"
Editado por F. A. Berry Jr.
McGraw Hill Book Company, Inc. 1945
- [9] U.S. Standard Atmosphere Supplements, 1966
- [10] U.S. Standard Atmosphere, 1976

[11] "Methods of experimental physics"

Volume 12 - Part B - Astrophysics - Radio Telescopes

2.2. Structure of the neutral atmosphere, by R. K. Crane

2.3. Atmospheric effects absorption and emission by atmospheric gasses, by J. W. Waters

Editado por M. L. Meeks

[12] J. Martín, J. Gómez-González, A. Barcia

"Simultaneous infrared spectral hygrometer and radiosonde measurements of the precipitable water vapour content in the atmosphere"

Infrared physics, Vol. 21, pp. 117-122, 1981

[13] J. L. Davis, T. A. Herring, I. I. Shapiro, A. E. E. Rogers,
G. Elgered

"Geodesy by radio interferometry: effects of atmospheric modelling errors on estimates of baseline length"

Radio Science, Vol. 20, No. 6, pp. 1593-1607

November-December, 1985

[14] G. Elgered

"Water vapor radiometry with applications to radio interferometry
and meteorology"

Technical Report No. 137, 1983

School of Electrical Engineering

Chalmers University of Technology, Göteborg, Sweden

[15] G. Elgered, B. Rönnäng, E. Winberg, J. Askne

"Satellite-Earth range measurements

I. Correction of the excess path length due to atmospheric water
vapour by ground based microwave radiometry"

Research Report No. 147, 1985

Research Laboratory of Electronics and Onsala Space Observatory

Chalmers University Of Technology, Göteborg, Sweden

[16] J. Askne, H. Nordins, E. Andersson, G. Elgered, B. Rönnäng

"Satellite-Earth range measurements

II. Modelling of tropospheric delay"

Research Report No. 155, 1985

Department of Radio and Space Science with Onsala Space
Observatory

Chalmers University of Technology, Göteborg, Sweden

[17] J. M. Johansson, G. Elgered, J. L. Davis

"Geodesy by radio interferometry: optimization of the wet path
delay algorithms using microwave radiometer data"

Research Report No. 152, 1987

Department of Radio and Space Science with Onsala Space
Observatory

Chalmers University of Technology, Göteborg, Sweden

[18] M. P. M. Hall

"Effects of the troposphere on radio communication"

I. E. E. E. Electromagnetics Waves

Series 8 / Peter Peregrins Ltd.

LISTADO DEL PROGRAMA

```

C-----FTN
C-----$FILES 0,2
      PROGRAM INREL
C-----
C-----CALCULA A PARTIR DE LOS PUNTOS NOTABLES DEL RADIOSONDEO
C-----LA CANTIDAD DE VAPOR DE AGUA PRECIPITABLE (EN MM) POR LOS
C-----METODOS DE HUMEDAD ESPECIFICA Y HUMEDAD ABSOLUTA.ASIMISMO
C-----DETERMINA LA OPACIDAD (EN NEPERS) DEL VAPOR DE AGUA , DEL
C-----OXIGENO Y LA EMISIVIDAD DE LA ATMOSFERA A DIFERENTES FRE
C-----CUENCIAS EN EL RANGO COMPRENDIDO ENTRE 10 Y 300 GHZ.
C-----TAMBIEN CALCULA EL RETARDO INTRODUCIDO POR LA ATMOSFERA,
C-----DIVIDIDO EN: HIDROSTATICO, POR VAPOR DE AGUA E IONOSFERICO
C-----
C-----TIENE DOS POSIBILIDADES, A ELECCION DEL USUARIO:
C-----
C-----INTEGRACION FINA (LENTA) QUE CONSISTE EN HACER INTERVALOS
C-----REGULARES DE 8 MILIBARES PARA INTREGRAR
C-----
C-----INTEGRACION RAPIDA (GRUESA): CONSISTE EN EL MISMO PROCESO QUE
C-----EN FINA, PERO SOLO HASTA EL PUNTO NOTABLE POR ENCIMA DE YEBES.
C-----DESPUES LA INTEGRAL SE REALIZA SOLAMENTE ENTRE PUNTOS NOTABLES.
C
C
      DIMENSION PN(50),TN(50),DN(50),HAB(50),P(500),T(500),E(500),
&ALT(500),EMI(50),TAUH2O(50),TAUO2(50),TAUT(50),IBUF(384),
&FRECU(50),RETION(50),EMIE(50),TUH2OE(50),TUO2E(50),
&TAUTE(50)
C-----
      DATA PI/3.1415926535898/
      DIMENSION IFE(4),IDIAS(4,10)
      CHARACTER DISKR*15,DISKD*15,IPROC*2,IPRO*2,IPRL*2
C--
      REAL IFREQI,IFREQF,INCRF
C--
      WRITE (*,*) 'OBSERVACION DESDE YEBES (SI,NO):'
      READ (*,'(A2)') IPRL
      IF (IPRL.EQ.'NO'.OR.IPRL.EQ.'no') THEN
        WRITE (*,*) 'ALTITUD DEL OBSERVATORIO (m):'
        READ (*,*) ALTYE
        WRITE (*,*) 'LATITUD DEL OBSERVATORIO (GRADOS):'
        READ (*,*) RLATIT
        RLATITUD=RLATIT*PI/180
      ELSE
        ALTRS=633.0
        ALTYE=920.0
        RLATIT=40.5234
        RLATITUD=0.7072670449
      ENDIF
C-----
      5 WRITE (*,*) 'DESEA PROCESAR NUEVOS DATOS (SI,NO):'
      READ (*,'(A2)') IPROC
      IF (IPROC.EQ.'NO'.OR.IPROC.EQ.'no') STOP
C-----INTEGRACION FINA O GRUESA
      WRITE (*,*) 'INTEGRACION FINA O GRUESA (0=GRUESA,1=FINA):'

```

```

      READ (*,*) ILENTO
C----FRECUENCIAS INICIAL,FINAL E INCREMENTO
      WRITE (*,*) 'FRECUENCIAS (INI., FIN., INCREM.):'
      READ (*,*) IFREQI,IFREQF,INCRF
C----ELEVACION DE LA ANTENA
182  WRITE (*,*) 'ELEVACION (ZENIT = 90):'
      READ (*,*) ELEVA
      IF ((ELEVA.LT.5).OR.(ELEVA.GT.90)) THEN
          WRITE(*,*)'ELEVACION NO VALIDA'
          GOTO 182
      ENDIF
      ELEV=ELEVA*PI/180
C----FICHERO DE RESULTADOS
      WRITE(*,*) 'FICHERO DE RESULTADOS: '
      READ(*,'(A)') DISKR
      OPEN(2,STATUS='NEW',FILE=DISKR,ERR=99,IOSTAT=IOS)
C----INTRODUCIR DATOS
      WRITE (*,*) 'DATOS DESDE EL TERMINAL (RS) : 1'
      WRITE (*,*) 'DATOS LEIDOS DE DISCO (RS) : 2'
      WRITE (*,*) 'DATOS DE ATMOSFERA ESTANDAR : 3'
      WRITE (*,*) 'TECLEAR LA OPCION ELEGIDA.'
      READ (*,'(I2)') ITE
      IF ((ITE.LT.3).AND.(IPRL.EQ.'NO'.OR.IPRL.EQ.'no')) THEN
          WRITE (*,*) 'ALTITUD DEL LUGAR DEL RADIOSONDEO (m):'
          READ (*,*) ALTRS
      ENDIF
C-----
      WRITE(2,281) ALTYE
      WRITE(2,282) RLATIT
      IF (ITE.LT.3) WRITE(2,283) ALTRS
281  FORMAT(' ALTITUD DEL OBSERVATORIO:',F7.1,' m')
282  FORMAT(' LATITUD DEL OBSERVATORIO:',F7.2,' GRADOS')
283  FORMAT(' ALTITUD DEL LUGAR DE LOS RADIOSONDEOS:',F7.1,' m')
      WRITE(2,284) ELEVA
284  FORMAT(' ELEVACION DE LA ANTENA: ',F7.2,' GRADOS')
C-----
      IF (ITE-2) 12,15,85
C
C----DATOS MANUALES (POR EL TERMINAL)
12  WRITE (*,*) 'FECHA,HORA (DIA,MES,ANNO,HORA: 01088300):'
      READ (*,'(4I2)') (IFE(I),I=1,4)
      WRITE(2,'(A,4I4)') ' FECHA, HORA:',(IFE(I),I=1,4)
      WRITE (*,*) 'NUMERO DE PUNTOS NOTABLES:'
      READ (*,*) N
      WRITE (*,*) 'PUNTO DE ROCIO (PR) O HUMEDAD RELATIVA (HR):'
      READ (*,'(A2)') IPRO
      IOPT=0
      IF (IPRO.EQ.'PR'.OR.IPRO.EQ.'pr') IOPT=1
      WRITE(2,*) 'PTO. NOTABLE: PRES.(mb) TEMP.(C) HUM.REL.(%)'
      WRITE(2,*) '-----'
      DO 9 I=1,N
      WRITE (*,8) I
8  FORMAT(I3,'.PRESION(MB) TEMPERATURA(C) HUM.REL(%) O PR(C):')
      READ (*,*) PN(I),TN(I),DN(I)

```



```

        WRITE (2,' (I6,A,F15.2,2F9.2)') I,'.-',PN(I),TN(I),DN(I)
9      CONTINUE
10     WRITE (*,*) 'NUMERO DEL PUNTO NOT. EQUIVOCADO (0=NINGUNO):'
        READ (*,*) IP
        IF(IP.EQ.0) GOTO 20
        WRITE (*,8) IP
        READ (*,*) PN(IP),TN(IP),DN(IP)
        write(2,' (I3,3F8.2,A)') IP,PN(IP),TN(IP),DN(IP),' **.correccion'
        GOTO 10

C
C-----LECTURA DE DATOS (ESCRITOS EN BINARIO) DESDE EL DISCO
C-----FICHERO DE DATOS
15     WRITE (*,*) 'NOMBRE DEL FICHERO DE DATOS (NOM:CS:CR:TI:TAM):'
        READ (*,' (A)') DISKD
        OPEN (55,STATUS='OLD',FILE=DISKD,ERR=99,IOSTAT=IOS)
C-----LOS DATOS EN DISCO CONTIENEN EL PUNTO DE ROCIO
        IOPT=1
C-----DIAS CONCRETOS A SELECCIONAR DESDE EL DISCO
        WRITE (*,*) 'NUM. DE DIAS CONCRETOS (0=NINGUNO):'
        NOPD=0
        READ (*,*) NDIAS
        IF(NDIAS.NE.0) THEN
            NOPD=1
            ICTD=0
            DO 16 I=1,NDIAS
                READ (*,' (4I2)') (IDIAS(J,I),J=1,4)
16     CONTINUE
        ENDIF
C-----LEE LOS DATOS DEL DISCO
17     READ (55,END=50,ERR=99,IOSTAT=IOS) (IFE(I),I=1,4),N
        READ (55,END=50,ERR=99,IOSTAT=IOS) (PN(I),TN(I),DN(I), I=1,N)
C-----DIAS EN CONCRETO
        IF( NOPD.EQ.1) THEN
            DO 18 I=1,NDIAS
                IF(IFE(1).EQ.IDIAS(1,I).AND.IFE(2).EQ.IDIAS(2,I).
&AND.IFE(3).EQ.IDIAS(3,I).AND.IFE(4).EQ.IDIAS(4,I)) THEN
                    ICTD=ICTD+1
                    GO TO 20
                ENDIF
18     CONTINUE
            GO TO 17
        ENDIF

C
C-----TRANSFORMA PUNTO DE ROCIO EN HUMEDAD RELATIVA (%)
20     IF (IOPT.NE.1) GO TO 31
        DO 25 I=1,N
            TIN=TN(I)+273.
            TD=DN(I)+273.
            IF (DN(I).EQ.52) THEN
                DN(I)=0
                GOTO 25
            END IF
            DN(I)=PREH2O(TD,12,12,1)*100./PREH2O(TIN,12,12,1)
            WRITE (*,24) I,DN(I)

```

```

24  FORMAT(1X,'HUMEDAD ',I3,F5.1)
25  CONTINUE
    GOTO 31

C
C-----ATMOSFERA ESTANDAR
85  N=40
    WRITE (*,*) 'TEMPERATURA Y HUMEDAD RELATIVA EN SUPERFICIE: 1'
    WRITE (*,*) 'TEMPERATURA EN SUPERFICIE UNICAMENTE           : 2'
    WRITE (*,*) 'TECLEAR LA OPCION ELEGIDA'
    READ (*,*) ITEM
    IF (ITEM-2) 121,125,125

C-----
121  WRITE (*,*) 'TEMPERATURA(C) Y HUMEDAD RELATIVA(%) EN SUPERFICIE:'
    READ (*,*) TS,HS
    write (2,221) TS,HS
221  format (' TEMPERATURA Y HUMEDAD RELATIVA EN SUPERFICIE:',
&F8.2,' C ',F8.2,' %'//)
    TSUP=TS+273
    ESUP=(HS/100)*PREH2O(TSUP,2,2,1)
    HABS=HUMABS(TSUP,ESUP)*EXP(ALTIE/2000.)
    WPREC=0.4*TS+3.3
    GOTO 128

C-----
125  WRITE (*,*) 'TEMPERATURA EN SUPERFICIE (C):'
    READ (*,*) TS
    write (2,224) TS
224  format (' TEMPERATURA EN SUPERFICIE:',F8.2,' C'//)
    WPREC=0.4*TS+3.3
    HABS=EXP(ALTIE/2000.)*WPREC/2

C-----
128  CONTINUE
    write(2,*)'      PRESION(mb) TEMPERATURA(C) HUMEDAD REL.(%)'
    write(2,*)'      -----'
    m=4
    DO 88 I=1,N
    RINCZ=ALTIE+(24000./N)*(I-1)
    IF (I.LE.17) THEN
        TN(I)=15-0.0065*RINCZ+(TS-9.02)
    ELSE
        TN(I)=-56.5+(TS-9.02)
    ENDIF
    PN(I)=1013.2*((1-0.0065*RINCZ/288.))**5.2568)
    ABSE= -RINCZ/2000.
    HAB(I)=HABS*EXP(ABSE)
    TNK=TN(I)+273
    E(I)=HAB(I)*TNK/216.9
    IF (E(I).GT.PREH2O(TNK,2,2,1)) THEN
        E(I)=PREH2O(TNK,2,2,1)
        WRITE(*,*) I,'.PRESION DE H2O > P. SATURACION'
    END IF
    DN(I)=(E(I)/PREH2O(TNK,2,2,1))*100
    if (M.EQ.4) then
        write (2,'(I3,A,F8.2,2F14.2)') I,'.-',PN(I),TN(I),DN(I)
    m=0

```

```

endif
m=m+1
88 CONTINUE
WRITE (*,122) WPREC
write (2,122) WPREC
122 FORMAT (' CANTIDAD DE AGUA PRECIPITABLE (RECTA): ',F6.2,' mm')
WRITE (*,143) HABS
write (2,143) HABS
143 FORMAT (' HUMEDAD ABSOLUTA: ',F8.2,' g/m3')
C-----
C-----INTERPOLACION LINEAL ENTRE PUNTOS NOTABLES CADA 8 MB PARA
C-----DETERMINAR TEMP. ALTURA Y PRESION PARCIAL DE VAPOR DE AGUA
C-----PUEDE SER VALIDO PARA TODOS LOS PUNTOS NOTABLES O SOLO HASTA
C-----EL PUNTO NOTABLE POR ENCIMA DE LA ALTURA DE YEBES,CASO DE
C-----INTEGRACION GRUESA
C-----
C-----CAMBIA LOS DATOS DE PRESION ERRONEOS
31 CONTINUE
DO 35 I=2,N
IF(PN(I-1).GT.100.AND.PN(I).LT.10) THEN
DO 36 IK=I,N
PN(IK)=PN(IK)*10.
36 CONTINUE
GOTO 32
END IF
35 CONTINUE
C-----DEFINE PUNTO NOTABLE A 20 MILIBARES
C-----
32 IF (PN(N).LT.20.) THEN
N=N+1
PN(N)=10
TN(N)=-70
DN(N)=0
END IF
C-----
C-----
ICHEQ=0
C-----
ICONT=1
P(1)=PN(1)
ALT(1)=ALTRS
IF (ITE.EQ.3) ALT(1)=ALTYE
T(1)=TN(1)+273.0
E(1)=PREH2O(T(1),DN(1),P(1),0)
DO 30 I=2,N
C-----
C-----INTEGRACION GRUESA: ICHEQ=1 A PARTIR DE ALTURA YEBES
IF (ICHEQ.EQ.1) GOTO 38
C-----
IF ((PN(I-1)-PN(I)).EQ.0) GOTO 130
AT=(TN(I-1)-TN(I))/(PN(I-1)-PN(I))
BT=TN(I)-AT*PN(I)
AD=(DN(I-1)-DN(I))/(PN(I-1)-PN(I))
BD=DN(I)-AD*PN(I)

```

```

      IPRE=(P (ICONT) -PN (I)) /8
      DO 28 IST=1, IPRE
C-----
C-----PRESION
      P (ICONT+1)=P (ICONT) -8.0
C-----TEMPERATURA
      T (ICONT+1)=P (ICONT+1) *AT+BT+273.0
C-----HUMEDAD RELATIVA
      HUMED=AD*P (ICONT+1) +BD
      E (ICONT+1)=PREH2O (T (ICONT+1) , HUMED, P (ICONT+1) , 0)
C-----ALTURA DEL PUNTO ICONT+1
      ALT (ICONT+1)=ALT (ICONT) +RINALT (P (ICONT) , T (ICONT) , E (ICONT) ,
&P (ICONT+1) , T (ICONT+1) , E (ICONT+1) )
      IF (ALT (ICONT) .LE. ALTYE .AND. ALT (ICONT+1) .GT. ALTYE) THEN
C-----
C-----INTEGRACION GRUESA: ILENTO=0 ==> ICHEQ=1
      IF (ILENTO.NE.1) ICHEQ=1
C-----
      INTE=ICONT
      DIFF1=ALTYE-ALT (INTE)
      DIFF2=ALT (INTE+1) -ALTYE
      IF (DIFF2.LT.DIFF1) INTE=INTE+1
      END IF
      ICONT=ICONT+1
28    CONTINUE
      GOTO 330
C-----
C-IL-SENTENCIAS PARA EL CASO DE INTEGRACION GRUESA
38    P (ICONT+1)=PN (I)
      T (ICONT+1)=TN (I) +273.0
      HUMED=DN (I)
      E (ICONT+1)=PREH2O (T (ICONT+1) , HUMED, P (ICONT+1) , 0)
C-----ALTURA DEL PUNTO ICONT+1
      ALT (ICONT+1)=ALT (ICONT) +RINALT (P (ICONT) , T (ICONT) , E (ICONT) ,
&P (ICONT+1) , T (ICONT+1) , E (ICONT+1) )
C-----
      ICONT=ICONT+1
330  CONTINUE
30    CONTINUE
C-----FUNCIONES DE MAPPING
      RMAPTA=1/SIN (ELEV)
      RA=0.001185*(1+0.6071E-4*(P (INTE) -1000) -0.1471E-3*E (INTE) +
&0.3072E-2*(T (INTE) -293) )
      RB=0.001144*(1+0.1164E-4*(P (INTE) -1000) +0.2795E-3*E (INTE) +
&0.3109E-2*(T (INTE) -293) )
      RC=-0.0090
      IF (ELEVA.LT.20) THEN
          RMAPES=1/(SIN (ELEV) +RA/(TAN (ELEV) +RB/(SIN (ELEV) +RC) ) )
      ELSE
          RMAPES=1/(SIN (ELEV) +RA/(SIN (ELEV) +RB/(SIN (ELEV) +RC) ) )
      ENDIF
C-----
C-----INTEGRACION DE LA CANTIDAD DE AGUA PRECIPITABLE (POR HUME-
C-----DAD ESPECIFICA Y ABSOLUTA). CONTRIBUCION A DIFERENTES FRE-
```

```

C-----CUENCIAS A LA OPACIDAD ATMOSFERICA POR EL H2O Y EL O2 Y LA EMI-
C-----SIVIDAD DE LA ATMOSFERA.
C-----
      WHES=0
      WHAB=0
      DO 34 I=1,50
      TAUH2O(I)=0
      TAUO2(I)=0
      EMI(I)=0
      TAUT(I)=0
34    CONTINUE
      HRINTE=E(INTE)*100/PREH2O(T(INTE),2,2,1)
      TINTE=T(INTE)-273
      WRITE(*,*)'ALTURA(m), TEMP.(C), Y HUM.REL.(%) EN EL OBSERVATORIO'
      WRITE(*,'(F8.2,2F12.2)')ALT(INTE),TINTE,HRINTE
      WRITE(2,*)'ALTURA(m), TEMP.(C), Y HUM.REL.(%) EN EL OBSERVATORIO'
      WRITE(2,'(F8.2,2F12.2)')ALT(INTE),TINTE,HRINTE
C-----
C-----INTEGRACION POR LA REGLA DE LOS TRAPECIOS
C-----
40    IONT=ICONT-1
      DO 42 I=INTE,IONT
C-----HUMEDAD ESPECIFICA
      WHES=WHES+5.08E-3*(P(I)-P(I+1))*(HUMES(P(I),E(I))+
&HUMES(P(I+1),E(I+1)))
C
C-----HUMEDAD ABSOLUTA
C      WHAB=WHAB+5E-4*(ALT(I+1)-ALT(I))*(HUMABS(T(I),E(I))+
C      &HUMABS(T(I+1),E(I+1)))
C
C-----OPACIDADES DEL H2O Y O2
      NPFR=1+(IFREQF-IFREQI)/INCRF
      DO 42 IE=1,NPFR
      FRECU(IE)=IFREQI+(IE-1)*INCRF
C
      TAUO2I=0.5*(ALT(I+1)-ALT(I))*100.0*
&(ABKO2(P(I),T(I),FRECU(IE))+ABKO2(P(I+1),T(I+1),FRECU(IE)))
C
      TAH2OI=0.5*(ALT(I+1)-ALT(I))*100.0*
&(ABKH2O(P(I),T(I),HUMABS(T(I),E(I)),FRECU(IE))+
&ABKH2O(P(I+1),T(I+1),HUMABS(T(I+1),E(I+1)),FRECU(IE)))
C
C-----EMISIVIDADES
      EMI1=TEMPR(T(I),FRECU(IE))*EXP(-TAUT(IE))
      EMI1=EMI1*(ABKO2(P(I),T(I),FRECU(IE))+ABKH2O(P(I),T(I),
&HUMABS(T(I),E(I)),FRECU(IE)))
C
C-----OPACIDADES EN EL PUNTO J+1
      TAUO2(IE)=TAUO2(IE)+TAUO2I
      TUO2E(IE)=TAUO2(IE)*RMAPTA
      TAUH2O(IE)=TAUH2O(IE)+TAH2OI
      TUH2OE(IE)=TAUH2O(IE)*RMAPTA
      TAUT(IE)=TAUH2O(IE)+TAUO2(IE)
      TAUTE(IE)=TAUT(IE)*RMAPTA

```

```

C
C-----EMISIVIDAD DE LA ATMOSFERA EN K
      EMI2=TEMPR(T(I+1),FRECU(IE))*EXP(-TAUT(IE))
      EMI2=EMI2*(ABKO2(P(I+1),T(I+1),FRECU(IE))+
&ABKH2O(P(I+1),T(I+1),HUMABS(T(I+1),E(I+1)),FRECU(IE)))
C
C
      EMI(IE)=EMI(IE)+0.5*(EMI1+EMI2)*(ALT(I+1)-ALT(I))*100.
C
42  CONTINUE
C
      DO 43 I=1,NPFR
      EMI(I)=EMI(I)/(1.-EXP(-TAUT(I)))
      EMIE(I)=EMI(I)*(1.-EXP(-TAUTE(I)))
43  CONTINUE
C
C      WRITE(*,43) WHAB
C44  FORMAT(' CANT. AGUA PREC. (HUM. ABSOLUTA):',F5.1,' mm'/)
      WRITE(*,45) WHES
      write(2,45) WHES
45  FORMAT(' CANT. AGUA PREC. (HUM. ESPECIFICA):',F5.1,' mm'/)
      WRITE(*,47)
      write(2,47)
      WRITE(*,417)
      write(2,417)
47  FORMAT(1X,'FRECS(GHz) OPACIDADES H2O,O2,TOTALES(NEP) EMISIV(K)
& T.BRILLO(K)')
417  FORMAT(1X,'-----')
& -----')
      DO 48 I=1,NPFR
      WRITE(*,61) FRECU(I),TUH2OE(I),TUO2E(I),TAUTE(I),EMI(I),EMIE(I)
      write(2,61) FRECU(I),TUH2OE(I),TUO2E(I),TAUTE(I),EMI(I),EMIE(I)
61  FORMAT(4X,F5.1,5X,3F8.3,5X,F7.1,3X,F7.1)
48  CONTINUE
      WRITE(*,*)
      WRITE(*,*)' EFECTO DE LA ATMOSFERA SOBRE LA FASE? (SI,NO): '
      READ(*,'(A2)')IPROC
      IF (IPROC.NE.'NO'.AND.IPROC.NE.'no') THEN
C
C---EN ESTA PARTE SE CALCULA EL RETARDO INTRODUCIDO POR LA ATMOSFERA,
C---TENIENDO EN CUENTA QUE ESTA COMPUESTO DE LAS SIGUIENTES CONTRI-
C---BUCIONES: RETARDO HIDROSTATICO, RETARDO DEBIDO AL VAPOR DE AGUA
C---Y RETARDO DEBIDO A LA IONOSFERA
C
C
C---RETARDO HIDROSTATICO ZENITAL
C
      RETS=0.2277*P(INTE)/(1-0.00266*COS(2*RLATITUD)-0.28E-6*ALTYE)
      RETS=RETS*RMAPES
C
C---RETARDO ZENITAL DEBIDO AL VAPOR DE AGUA
C
      PARCRET=0
      DO 530 I=INTE,IONT

```

```

SUBIN1=E(I)/(T(I)**2)
SUBIN2=E(I+1)/(T(I+1)**2)
PARCRET=PARCRET+0.5*(ALT(I+1)-ALT(I))*(SUBIN1+SUBIN2)
530 CONTINUE
RETH=37.7*PARCRET
RETH=RETH*RMAPES
C
C---RETARDO MAXIMO DEBIDO A LA IONOSFERA (FREC EN GHZ)
C
DO 540 I=1,NPFR
RETIION(I)=3900/(FRECU(I)**2)
RETIION(I)=RETIION(I)*RMAPTA
540 CONTINUE
C
C---PRESENTACION
C
WRITE(*,551)RETS
WRITE(2,551)RETS
WRITE(*,561)RETH,RETH/30
WRITE(2,561)RETH,RETH/30
WRITE(*,571)RETS+RETH
WRITE(2,571)RETS+RETH
551 FORMAT(/' RETARDO HIDROSTATICO: ',F7.1,' cm')
561 FORMAT(' RETARDO DEBIDO AL VAPOR DE AGUA: ',F5.1,' cm = ',
&F6.2,' ns')
571 FORMAT(' RETARDO TOTAL NO DISPERSIVO: ',F7.1,' cm'/)
WRITE(*,*)
&' FRECS(GHZ) R.IONOSFERICO(cm) R.TOTAL FASE(rad) R.TOTAL(ns)'
WRITE(2,*)
&' FRECS(GHZ) R.IONOSFERICO(cm) R.TOTAL FASE(rad) R.TOTAL(ns)'
WRITE(*,*)
&' -----'
WRITE(2,*)
&' -----'
DO 550 I=1,NPFR
RETT=RETIION(I)+RETS+RETH
RETFASE=(2*PI/30)*FRECU(I)*RETT
RETTIEM=RETT/30
WRITE(*,581)FRECU(I),RETIION(I),RETFASE,RETTIEM
WRITE(2,581)FRECU(I),RETIION(I),RETFASE,RETTIEM
550 CONTINUE
581 FORMAT(4X,F5.1,F14.1,F20.2,F14.2)
ENDIF
close(2,STATUS='KEEP',ERR=99,IOSTAT=IOS)
IF(ITE.NE.2) GOTO 5
IF(ICTD.EQ.NDIAS.AND.NOPD.EQ.1) GOTO 50
GOTO 17
C----CIERRA EL FICHERO DE DATOS
50 CLOSE(55,STATUS='KEEP',ERR=99,IOSTAT=IOS)
GOTO 5
99 WRITE(*,101) IOS
101 FORMAT(' ERROR LECTURA DISCO=',I5)
130 WRITE(*,*) 'ERROR: DOS PUNTOS CON IGUAL PRESION.'
STOP

```

```

      END
C-----
C-----#####-----
C
      FUNCTION PREH2O(T,H,P,I)
C
C-----PRESION DE VAPOR DE AGUA. I=1 PRESION DE SATURACION A TEM-
C-----PERATURA T.I=0 PRESION DE VAPOR DE AGUA A P (MB),T(K) Y
C-----HUMEDAD RELATIVA H (%).
C
      PREH2O=25.22*(T-273)/T
      PREH2O=PREH2O-5.31*ALOG(T/273)
      PREH2O=6.105*EXP(PREH2O)
      IF(I.EQ.1) RETURN
      AIN=(1-(1-H/100.)*PREH2O/P)*100.0
      PREH2O=PREH2O*H/AIN
      RETURN
      END
C-----
C
      FUNCTION RINALT(P1,T1,E1,P2,T2,E2)
C
C-----INCREMENTO EN ALTURA (EN METROS) ENTRE PUNTO 1 Y 2
C
      S1=T1*(1+.388*E1/P1)
      S2=T2*(1+.388*E2/P2)
      RINALT=14.645*(S1+S2)*ALOG(P1/P2)
      RETURN
      END
C-----
C
      FUNCTION HUMABS(T,E)
C
C-----HUMEDAD ABSOLUTA EN GRAMOS/M**3.T=TEMPERATURA EN K Y E
C-----PRESION PARCIAL DEL VAPOR DE AGUA EN MB
C
      HUMABS=216.9*E/T
      RETURN
      END
C-----
C
      FUNCTION HUMES(P,E)
C
C-----HUMEDAD ESPECIFICA (UNIDADES IGUAL QUE EN CASOS ANTERIORES)
C
      HUMES=622.*E/(P-0.378*E)
      RETURN
      END
C-----
C
      FUNCTION ABK02(PREASU,TEMP,FREQ)
C
C
C THIS FUNCTION COMPUTES THE O2 ABSORPTION COEFFICIEN-

```



```

C   TE IN CM**-1 AT A GIVEN FREQUENCY IN THE RANGE FROM
C   10 TO 300 GHZ.PREASURE AND TEMPERATURE ARE THE ONLY
C   METERELOGICAL PARAMETERS NEEDED. ALL O2 TRANSITIONS
C   BETWEEN 20GHZ AND 300 GHZ AND THE KINETIC LINE SHAPE
C   ARE USED IN THE CALCULATIONS.
C
C
C
C-----UNITS:PREASURE:MILLIBARS,TEMPERATURE:KELVINS,FRE-
C           QUENCY:GHZ.
C
C
C           DIMENSION FREQP(20),FREQN(20)
C           REAL N
C-----FREQP AND FREQN REPRESENT THE LINE FREQUENCY FOR
C           DELTJ=+1 AND DELTJ=-1 TRANSITIONS.
C
C           DATA FREQP/56.264766,58.44658,59.590978,60.434776,
C           <61.15057,61.800169,62.411223,62.997991,63.568520,
C           <64.1277777,64.678914,65.22412,65.764744,66.30206,
C           <66.83677,67.36951,67.90073,68.4308,68.9601,69.4887/
C           DATA FREQN/118.750343,62.486255,60.306044,
C           <59.164215,58.323885,57.612488,56.968180,56.363393,
C           <55.783819,55.221372,54.671145,54.1302,53.5959,
C           <53.0669,52.5424,52.0214,51.50302,50.9873,50.4736,
C           <49.9618/
C           DATA PI/3.1415926535898/
C
C
C           C1=1.44E-5*PREASU*FREQ/(TEMP**3.0)
C
C-----SET THE ABSORPTION COEFFICIENT TO ZERO
C           ABKO2=0
C
C-----COMPUTE LINE WIDTH FOR 1- AT 118GHZ(FREQN(1))
C           DELF1=0.00141*PREASU*300.0/TEMP
C
C-----COMPUTE LINEWIDTH FOR NON-RESONANT TRANSITION AND
C-----TAKE INTO THE ACCOUNT LINE OVERLAPPING (EMPIRICAL
C           LINEWIDTH
C           DELF2=0.035133+DELF1/3.0
C-----COMPUTE LINE PROFILE FOR NON-RESONANT LINES
C           PROFNR=2.0*FREQ*DELF2/(PI*(FREQ**2+DELF2**2))
C
C-----COMPUTE FOR EACH O2 TRANSITION ABKO2
C           DO 100 I=1,20
C
C-----COMPUTE LINE PROFILE FOR N=-1 (WHEN N=1)
C           IF(I.EQ.1) PROFN=RLINR(FREQ,DELF1,FREQN(I))
C           DELFRE=DELF1
C           IF(DELF1.GE.0.0527) DELFRE=DELF2
C
C-----COMPUTE LINE PROFILE FOR N=1+
C           PROFP=RLINR(FREQ,DELFRE,FREQP(I))

```

```

C
C----COMPUTE LINE PROFILE FOR N=1-, EXCEP FOR LINE AT 118GHZ
      IF (I.EQ.1) GOTO 120
      PROFN=RLINR(FREQ, DELFRE, FREQN(I))
120  CONTINUE

```

```

C
C----STATISTICAL WEIGHTS*MATRIX ELEMENTS FOR EACH TRANSITION
      N=2*I-1
      PRODN=(N+1)*(2*N-1)/N
      PRODP=N*(2*N+3)/(N+1)
      PRODNR=2*(N**2+N+1)*(2*N+1)/(N*(N+1))

```

```

C----ENERGY LEVEL
      ENERN=2.07*N*(N+1)/TEMP
      FACT=EXP(-ENERN)

```

```

C
C----COMPUTE ABSORPTION COEFFICIENT
      ABKO2=ABKO2+C1*(PRODP*PROFP*FREQP(I)+PRODN*PROFN*
      <FREQN(I)+PRODNR*PROFN/2.0)*FACT
100  CONTINUE
      RETURN
      END

```

```

C-----

```

```

C
      FUNCTION ABKH2O (PREASU, TEMP, DENH2O, FREQ)

```

```

C
C      THIS SUBROUTINE COMPUTES THE ABSORPTION COEFFICIENT OF
C      H2O AT A GIVEN FREQUENCY IN THE RANGE FROM 1 GHZ TO
C      1000 GHZ. THE METEOROLOGICAL CONDITION NEEDED ARE PREA-
C      SURE, TEMPERATURE AND WATER VAPOR CONCENTRACION. ALL
C      TRANSITIONS BETWEEN 20GHZ AND 1000 GHZ ARE INCLUDED
C      IN THE CALCULATIONS (PEPE'S PAR.). THE KINETIC LINE PROFILE AND THE
C      ADDICIONAL CORRECTION AS DECRIBED BY J.W. WATERS IN
C      "METHODS OF EXPERIMENTAL PHYSICS" ARE USED.

```

```

C
C      UNITS: PREASURE IN MILLIBARS, TEMPERATURE IN KELVINS, WATER
C      VAPOR CONCENTRACION IN GRAMS/METER CUBED AND FRE-
C      QUENCY IN GHZ.

```

```

C
      DIMENSION FREQTR(19), RMATRX(19), WEIGHT(19), ENERL(19),
      <DELTF0(19), DELH2O(19), XPARAM(19)

```

```

C
      DATA FREQTR/22.23515, 183.31012, 323.0, 325.1538, 380.1968, 390.0,
      <436.0, 438.0, 442.0, 448.0008, 470.8889, 474.69, 488.49, 556.936,
      <620.7, 752.03, 916.62, 970.31, 987.94/
      DATA WEIGHT/3., 1., 3., 1., 3., 1., 1., 3., 3., 3., 1., 1., 1., 3., 3.,
      <1., 1., 1., 1./
      DATA RMATRX/0.057, 0.102, 0.089, 0.091, 0.123, 0.068, 0.088, 0.101,
      <0.088, 0.132, .102, .118, .036, 1.5, .122, 2.073, .161, .262, .756/
      DATA ENERL/446.56, 136.16, 1283.02, 315.78, 212.16, 1525.31, 1045.03,
      <742.11, 1045.11, 285.42, 742.074, 488.135, 586.48, 23.794, 488.108,
      <70.091, 285.217, 383.837, 37.137/
      DATA DELTF0/2.85, 2.68, 2.3, 3.03, 3.19, 2.11, 1.50, 1.94, 1.51, 2.47,

```

```

<1.89,2.07,2.58,3.33,2.28,3.13,2.59,2.48,3.09/
  DATA DELH2O/13.68,14.49,12.04,15.21,15.84,11.42,7.94,
<10.44,8.13,14.24,10.56,11.95,14.77,14.66,12.78,13.93,14.06,
<14.16,15.2/
  DATA XPARAM/0.626,0.649,0.42,0.619,0.63,0.33,0.29,0.36,0.332,0.51,
<.38,.38,.57,.645,.6,.69,.676,.56,.66/

```

C
C

```

CONST1=1.44*DENH2O*FREQ/(TEMP**1.5)
FACT=DENH2O*TEMP/PREASU

```

C

```

C-----SET ABSORPTION COEFFICIENT,ABKH2O,TO 0
  ABKH2O=0
  DO 100 I=1,19

```

C

```

C----COMPUTE DELTA FREQ. FOR TRANSITION I
  FACTO=(PREASU/1013.)*DELTF0(I)*((300./TEMP)**XPARAM(I))
  FACTOR=(DELH2O(I)/DELTF0(I))-1
  DELFRE=FACTO*(1+0.0046*FACT*FACTOR)

```

C

```

C-----COMPUTE LINE PROFILE
  PROF=RLINR(FREQ,DELFRE,FREQTR(I))

```

C

```

C-----CONVERT ENERGIES IN CM**-1 TO KELVIN
  EM=(ENERL(I)+FREQTR(I)/29.97925)*1.439/TEMP
  EL=ENERL(I)*1.439/TEMP

```

C

```

C----COMPUTE ABSORPTION COEFFICIENT
  ABKH2O=ABKH2O+CONST1*(EXP(-EL)-EXP(-EM))*WEIGHT(I)*RMATRX(I)*PROF
  100 CONTINUE

```

C

```

C----- ADD THE ADDICIONAL CORRECTION
  C1=1.08E-11*DENH2O*((300./TEMP)**2.1)
  ABKH2O=ABKH2O+C1*(PREASU/1000.0)*(FREQ**2)
  RETURN
  END

```

C

C

```

FUNCTION RLINR (FREQ,DELFRE,FREQTR)

```

C

```

C THIS FUNCTION COMPUTES THE KINETIC LINE PROFILE AT FREQUEN-
C CY,FREQ,FOR THE LINE WIDTH,DELFRE.ALL INPUT PARAMETERS MUST
C BE IN THE SAME UNITS.IF THE INPUT UNITS ARE IN GHZ THE LINE
C PROFILE IS IN GHZ**(-1)

```

C

```

  DATA PI/3.1415926535898/

```

```

  DIF1=(FREQTR**2-FREQ**2)**2
  FACT2=4.0*(FREQ**2)*(DELFRE**2)
  FACT1=4.0*FREQTR*FREQ*DELFRE/PI
  RLINR=FACT1/(DIF1+FACT2)
  RETURN
  END

```

C

C

```
FUNCTION TEMPR (TEM,FREQ)
C---ESTA FUNCION CALCULA LA TEMPERATURA DE RADIACION A LA FRECUEN-
C---CIA FREQ (GHz) A LA TEMPERATURA ABSOLUTA TEM (K)
C
  CTE=48E-3*FREQ
  TEMPR=CTE/ (EXP (CTE/TEM) -1)
  RETURN
  END
```

C-----

